

Технология компонентного программирования

Лабораторная работа №1

Delphi, Paradox и Excel

Цель занятия – знакомство с методами взаимодействия в среде Delphi с приложением MS Excel – в данной работе конкретно научиться импортировать Excel-таблицу в файл БД типа Paradox. Вторая часть – экспорт Paradox-файла в Excel-файл.

Возьмём, например, такую таблицу:

	A	B	C	D	E	F	G	
1			Список пролеченных в стационаре					
2			2019 год					
3								
4	№ п/п	месяц	Ф.И.О.	Дата рождения	нач. леч.	конец леч.	Сумма	
5	1	2	Петров Иван Ильич	21.02.1970	11.02.19	15.02.19	1 234,50	
6	2	2	Иванов Пётр Семёнович	13.11.1950	12.02.19	18.02.19	3 245,00	
7	3	3	Воробьёв Кузьма Фролович	31.12.1920	13.03.19	23.03.19	5 450,00	
8								

Тогда таблица Paradox будет иметь следующую структуру:


Таблица **Primer:**

Назначение поля	Название поля	Тип	Размер
№ п/п	NPP	Integer	
Месяц	Mes	SmallInt	
Ф. И. О.	FIO	String	60
Дата рождения	DR	DateTime	
Дата нач. леч.	Dat1	DateTime	
Дата оконч. леч.	Dat2	DateTime	
Сумма	Sum	Currency	

Для работы с Excel надо подключить следующие модули:

ExcelXP, ComObj.

Интерфейс же надо бы сделать примерно таким:



Поле для отображения импортируемого Excel-файл – это компонент LabeledEdit.

Главная кнопка, по которой вызывается диалоговое окно (компонент OpenFileDialog) и задаётся файл – это SpeedButton. С остальными компонентами вы знакомы.

Весь процесс импорта мы запустим одной кнопкой – через событие SpeedButton1.OnClick – как только будет задан файл, сразу же начинается его чтение и загрузка в таблицу Paradox, которую будем создавать каждый раз заново.

implementation

```
{$R *.dfm}
```

var

```
Excel: OleVariant; // Excel-приложение
ExcelWB: OleVariant; // Excel-книга
ExcelWS: OleVariant; // Excel-Лист
ExcelOpened: boolean;
```

```
procedure TForm1.SpeedButton1Click(Sender: TObject);
```

const

```
FName = 'Primer.db';
```

var

```
s : string;
n,r: integer;
```

begin

```
if not OpenFileDialog1.Execute then exit;
LabeledEdit1.Text:= OpenFileDialog1.FileName;
if FileExists(FName) then DeleteFile(FName);
DataSource1.DataSet:= nil; // пока отключаем связь с DBGrid1
// создаём файл Paradox для приёма записей из Excel
```

Здесь же создаём файл Paradox для приёма записей из Excel (это вы уже умеете сами), открываем Table1, подключаем Excel и загружаем записи в Paradox-файл.

```

    // Подключаем Excel
ExcelOpened:= false;
try
    Excel:= GetActiveOleObject('Excel.Application');
    ExcelOpened:= true;
except
    Excel:= CreateOleObject('Excel.Application');
end;
Excel.Visible:=false; // Убираем с глаз Excel
ExcelWB:= Excel.Workbooks.Open(LabeledEdit1.Text);
ExcelWS:= ExcelWB.WorkSheets.Item[1];

r:= 5; // начальная строка с данными
n:= ExcelWS.Cells[r,1];
while n > 0 do begin
    Table1.Append;
    Table1.Fields[0].Value:= n;
    Table1.Fields[1].AsInteger:= ExcelWS.Cells[r,2];
    Table1.Fields[2].AsString:= ExcelWS.Cells[r,3];
    s:= ExcelWS.Cells[r,4];
    Table1.Fields[3].Value:= StrToDate(s);
    s:= ExcelWS.Cells[r,5];
    Table1.Fields[4].Value:= StrToDate(s);
    s:= ExcelWS.Cells[r,6];
    Table1.Fields[5].Value:= StrToDate(s);
    s:= ExcelWS.Cells[r,7];
    Table1.Fields[6].Value:= StrToFloat(s);
    Table1.Post;
    inc(r);
    n:= ExcelWS.Cells[r,1];
end;
Table1.Close;

if ExcelOpened then // если Excel был загружен, то возвращаем его
    Excel.Visible:= true
else
    Excel.Quit; // если Excel не был загружен, то закрываем его

DataSource1.DataSet:= Table1; // включаем связь с DBGrid1
Table1.Open;

```

На этом первую часть закончили – Excel-файл загружен.

Вторая часть задачи – выгрузка текущего набора данных в Excel-файл.

Для усвоения более широкого круга приёмов работы с Excel-приложением добавим в работу различные операции форматирования Excel-таблицы, включая отображения сетки в тех строках, где имеется информация. За это будет отвечать процедура Setka.

```
{ ----- Выгрузка в Excel ----- }

procedure TForm1.BitBtn1Click(Sender: TObject);
var
  FName, ext, BM: string;
  ExcRange: OleVariant;
  i: integer;

procedure Setka (R1, R2: string);
begin
  ExcRange:= ExcelWS.Range [R1, R2];
  ExcRange.Borders[xlEdgeLeft].LineStyle := xlContinuous;
  ExcRange.Borders[xlEdgeLeft].Weight := xlMedium;
  ExcRange.Borders[xlEdgeTop].LineStyle := xlContinuous;
  ExcRange.Borders[xlEdgeTop].Weight := xlMedium;
  ExcRange.Borders[xlEdgeBottom].LineStyle := xlContinuous;
  ExcRange.Borders[xlEdgeBottom].Weight := xlMedium;
  ExcRange.Borders[xlEdgeRight].LineStyle := xlContinuous;
  ExcRange.Borders[xlEdgeRight].Weight := xlMedium;
  ExcRange.Borders[xlInsideVertical].LineStyle := xlContinuous;
  try
    ExcRange.Borders[xlInsideHorizontal].LineStyle := xlContinuous;
  except
    // skip
  end;
  ExcRange.Font.Size := 10;
  ExcRange.WrapText := true;
  ExcRange.VerticalAlignment := xlCenter;
end;

begin {BitBtn1Click}
  if not Table1.Active then exit;
  if Table1.RecordCount = 0 then exit;
  if not SaveDialog1.Execute then exit;
  FName:= SaveDialog1.FileName;
  ext:= LowerCase(ExtractFileExt(FName));
  if (ext <> '.xls') and (ext <> '.xlsx') then
    FName:= ChangeFileExt(FName, '.xls');
    // Подключаем Excel
  ExcelOpened:=false;
  try
    Excel:= GetActiveOleObject('Excel.Application');
    ExcelOpened:= true;
  except
    Excel:= CreateOleObject('Excel.Application');
  end;
  Excel.Visible:= false; // Убираем с глаз Excel
  ExcelWB:= Excel.Workbooks.Add;
  ExcelWS:= ExcelWB.WorkSheets.Item[1];
```

```

    // работаем с Excel-файлом...
    // Заголовок таблицы...
ExcRange:= ExcelWS.Range['B1','E1'];
ExcRange.HorizontalAlignment:= xlCenter;
ExcRange.VerticalAlignment:= xlCenter;
ExcRange.Merge;
ExcRange.Font.Name:= 'Arial';
ExcRange.Font.Size:= 12;
ExcRange.Font.Bold:= True;
ExcelWS.Cells[1,2]:= 'Список пролеченных в стационаре';
ExcelWS.Cells[2,3]:= '2020 год';
ExcelWS.Cells[2,3].Font.Bold:= true;
ExcelWS.Cells[2,3].HorizontalAlignment:= xlRight;
    // Названия столбцов...
ExcelWS.Columns[1].ColumnWidth:= 5.11;
ExcelWS.Columns[2].ColumnWidth:= 5.89;
ExcelWS.Columns[3].ColumnWidth:= 27.00;
ExcelWS.Columns[4].ColumnWidth:= 10.00;
ExcelWS.Columns[5].ColumnWidth:= 10.00;
ExcelWS.Columns[6].ColumnWidth:= 10.00;
ExcelWS.Columns[7].ColumnWidth:= 10.56;
ExcelWS.Rows[4].RowHeight:= 26.4;
ExcelWS.Cells[4,1]:= '№ п/п';
ExcelWS.Cells[4,2]:= 'Месяц';
ExcelWS.Cells[4,3]:= 'Ф.И.О.';
ExcelWS.Cells[4,4]:= 'Дата рождения';
ExcelWS.Cells[4,5]:= 'нач. леч.';
ExcelWS.Cells[4,6]:= 'конец леч.';
ExcelWS.Cells[4,7]:= 'Сумма';
Setka('A4','G4');
ExcRange:= ExcelWS.Range['A4','G4'];
ExcRange.HorizontalAlignment:= xlCenter;
ExcRange.VerticalAlignment:= xlCenter;
ExcRange.WrapText:= True;
ExcRange.Font.Name:= 'Arial';
ExcRange.Font.Size:= 10;
ExcelWS.Columns[2].HorizontalAlignment:= xlCenter;
ExcelWS.Columns[7].NumberFormat:= '# ##0,00';

    // заполнение таблицы данными
DataSource1.DataSet.DisableControls; // бездействие связи
i:= 5;
with Table1 do begin
    BM:= Bookmark;
    First;
    while not Eof do begin
        ExcelWS.Cells[i,1]:= Fields[0].AsInteger; // NPF
        ExcelWS.Cells[i,2]:= Fields[1].AsInteger; // Mes
        ExcelWS.Cells[i,3]:= Fields[2].AsString; // FIO
        ExcelWS.Cells[i,4]:= Fields[3].AsString; // DR
        ExcelWS.Cells[i,5]:= Fields[4].AsString; // Dat1
        ExcelWS.Cells[i,6]:= Fields[5].AsString; // Dat2
        ExcelWS.Cells[i,7]:= Fields[6].AsCurrency; // Sum
        inc(i);
    Next
    end;
    Bookmark:= BM;
end;
Setka('A5','G' + IntToStr(i-1));
ExcelWS.Cells[i,6]:= 'Итого';
ExcelWS.Cells[i,7].Formula:= '=SUM(G5:G' + IntToStr(i-1) + ')';

```


```

ExcelWS.Cells[i,6].Font.Bold:= true;
ExcelWS.Cells[i,7].Font.Bold:= true;

DataSource1.DataSet.EnableControls; // ОЖИВЛЯЕМ СВЯЗЬ
Excel.ActiveWorkbook.SaveAs(FileName:= FName, FileFormat:= xlNormal);
ShowMessage('Файл готов');
    // показываем результат
Excel.Visible:=true;
end; {BitBtn1Click}

```

Окончательный вид программы



Полученный Excel-файл

	A	B	C	D	E	F	G	
1		Список пролеченных в стационаре						
2		2020 год						
3								
4	№ п/п	Месяц	Ф.И.О.	Дата рождения	нач. леч.	конец леч.	Сумма	
5	1	2	Петров Иван Ильич	21.02.1970	11.02.2019	15.02.2019	1 234,50	
6	2	2	Иванов Пётр Семёнович	13.11.1950	12.02.2019	18.02.2019	3 245,00	
7	3	3	Воробьёв Кузьма Фролович	31.12.1920	13.03.2019	23.03.2019	1 200,00	
8						Итого	5 679,50	
9								